# Finding series coefficients

Given

b[0] := 1;
Sum[Binomial[n, k]*((2*n - 2*k - 1)!!)^2*b[k + 1], {k, 0, n}] == ((2*n + 1)!!)^2;


is there a way to find the coefficients b[n] using InverseSeries, SeriesCoefficient, or some other method?

• Do you need a closed form? Jan 24 at 4:09

If you don't need a closed form and merely need a certain number of terms, you can just solve it directly, like so:

Choose a certain number of terms, say

nTotal = 4;


Then,

sum1 = Table[
Sum[Binomial[n, k]*((2*n - 2*k - 1)!!)^2*b[k + 1], {k, 0, n}],
{n, 0, nTotal}]
sum2 = Table[((2*n + 1)!!)^2, {n, 0, nTotal}]
First@Solve[sum1 == sum2 // Thread, Array[b, nTotal + 1]]
(* {b[1], b[1] + b[2], 9 b[1] + 2 b[2] + b[3],
225 b[1] + 27 b[2] + 3 b[3] + b[4],
11025 b[1] + 900 b[2] + 54 b[3] + 4 b[4] + b[5]} *)
(* {1, 9, 225, 11025, 893025} *)
(* {b[1] -> 1, b[2] -> 8, b[3] -> 200, b[4] -> 9984, b[5] -> 824064} *)


This is a lower triangular system, so one can use LinearSolve[] with SparseArray[] for this:

With[{nTotal = 9},
LinearSolve[SparseArray[{n_, k_} /; n >= k :>
Binomial[n - 1, k - 1] ((2 n - 2 k - 1)!!)^2,
{nTotal, nTotal}],
Table[((2 n + 1)!!)^2, {n, 0, nTotal - 1}]]]
{1, 8, 200, 9984, 824064, 101253120, 17313776640, 3930091683840, 1143354433536000}